



UNIVERSIDADE FEDERAL DO RIO DE JANEIRO
ESCOLA DE QUÍMICA



EQE776 Modelagem e Simulação de Processos

Aula 08. Reator de Van der Vusse em EMSO

Professor: Roymel Rodríguez Carpio

E-mail: roymel@eq.ufrj.br

Recapitulando

- Modelo customizado de separador por membrana em Aspen Plus

Temas da aula

- Reator de Van der Vusse em EMSO

Apresentação do problema

- O reator de Van der Vusse é um exemplo clássico em engenharia química, muito usado para estudar cinética complexa e problemas de otimização em reatores contínuos.
- Ele descreve um sistema de reações, que apresenta reações em série e em paralelo, dentro de um reator contínuo de mistura perfeita (CSTR).
- É um exemplo didático que envolve competição entre reações e formação de subprodutos.

Apresentação do problema

Sistema de reações:

$A \xrightarrow{k_1} B$: Reação desejada

$B \xrightarrow{k_2} C$: Reação indesejada em série

$2A \xrightarrow{k_3} D$: Reação indesejada em paralelo

Onde:

A: reagente

B: produto desejado

C e D: subprodutos indesejados

Apresentação do problema

Modelo do sistema:

$$V \frac{dC_a}{dt} = Q_{in}C_a^{in} - Q_{out}C_a^{out} - r_1V - 2r_3V$$

$$r_1 = k_1C_a$$

$$r_2 = k_2C_b$$

$$V \frac{dC_b}{dt} = Q_{in}C_b^{in} - Q_{out}C_b^{out} + r_1V - r_2V$$

$$r_3 = k_3C_a^2$$

$$\tau = \frac{V}{Q_{in}}$$

$$V \frac{dC_c}{dt} = Q_{in}C_c^{in} - Q_{out}C_c^{out} + r_2V$$

$$Q_{out} = Q_{in}$$

$$C_i^{out} = C_i$$

$$V \frac{dC_d}{dt} = Q_{in}C_d^{in} - Q_{out}C_d^{out} + r_3V$$

$$P_{out} = P_{in}$$

$$T_{out} = T_{in}$$

Apresentação do problema


Dados:

$$C_a^{in} = 0,05 \frac{\text{mol}}{\text{L}}$$

$$Q_{in} = 250 \frac{\text{L}}{\text{min}}$$

$$T_{in} = 550 \text{ K}$$

$$P_{in} = 3 \text{ bar}$$



$$k_1 = 0,5 \text{ min}^{-1}$$

$$k_2 = 0,2 \text{ min}^{-1}$$

$$k_2 = 0,2 \text{ min}^{-1} \cdot \text{L} \cdot \text{mol}^{-1}$$

$$\begin{array}{cc} Q_{out} & \\ C_a^{out} & C_b^{out} \\ C_c^{out} & C_d^{out} \\ T_{out} & P_{out} \end{array}$$

$$V = 1000 \text{ L}$$

$$C_a^{t=0} = 0,05 \frac{\text{mol}}{\text{L}}$$

Criação dos modelos

Modelo de corrente de matéria

```
5 Model MaterialStream
6
7 VARIABLES
8
9 T      as positive (Brief="Temperature", Unit='K');
10 P      as positive (Brief="Pressure", Unit='bar');
11 Fvol    as positive (Brief="Volumetric Flow", Unit='l/min');
12 Ca      as positive (Brief = "Molar concentration component A", Unit='mol/l');
13 Cb      as positive (Brief = "Molar concentration component B", Unit='mol/l');
14 Cc      as positive (Brief = "Molar concentration component C", Unit='mol/l');
15 Cd      as positive (Brief = "Molar concentration component D", Unit='mol/l');
16
17 end
```


Criação dos modelos

Modelo de fonte e sumidouro

```
20 Model MaterialSource
21
22     VARIABLES
23
24     out Outlet as MaterialStream (Brief="Outlet Stream");
25
26 end
27
28
29 Model MaterialSink
30
31     VARIABLES
32
33     in Inlet as MaterialStream (Brief="Inlet Stream");
34
35 end
```

Criação dos modelos

Modelo de reator de Van der Vusse (parâmetros e variáveis)

```
38 Model cstr_vdv
39
40 PARAMETERS
41
42 k1 as positive (Brief="Reaction rate constant", Unit='1/min');
43 k2 as positive (Brief="Reaction rate constant", Unit='1/min');
44 k3 as positive (Brief="Reaction rate constant", Unit='(1/min)*(1/mol)');
45 V as positive (Brief="Reactor volume", Unit='l');
46
47 VARIABLES
48 in Inlet as MaterialStream (Brief="Inlet stream");
49 out Outlet as MaterialStream (Brief="Outlet stream");
50
51 Ca as positive (Brief="Molar concentration component A", Unit='mol/l');
52 Cb as positive (Brief="Molar concentration component B", Unit='mol/l');
53 Cc as positive (Brief="Molar concentration component C", Unit='mol/l');
54 Cd as positive (Brief="Molar concentration component D", Unit='mol/l');
55 tau as positive (Brief="Reactor mean residence time", Unit='min');
56 r1 as positive (Brief="Reaction rate", Unit='mol/(min*l)');
57 r2 as positive (Brief="Reaction rate", Unit='mol/(min*l)');
58 r3 as positive (Brief="Reaction rate", Unit='mol/(min*l)');
```

Criação dos modelos

Modelo de reator de Van der Vusse (equações, 1ra parte)

```
60 EQUATIONS
61
62 "Reaction Rate - First Reaction"
63 r1 = k1*Ca;
64
65 "Reaction Rate - Second Reaction"
66 r2 = k2*Cb;
67
68 "Reaction Rate - Third Reaction"
69 r3 = k3*Ca^2;
70
71 "A Component Mass Balance"
72 v*diff(Ca) = Inlet.Fvol*Inlet.Ca - Outlet.Fvol*Outlet.Ca - r1*v - 2*r3*v;
73
74 "B Component Mass Balance"
75 v*diff(Cb) = Inlet.Fvol*Inlet.Cb - Outlet.Fvol*Outlet.Cb + r1*v - r2*v;
76
77 "C Component Mass Balance"
78 v*diff(Cc) = Inlet.Fvol*Inlet.Cc - Outlet.Fvol*Outlet.Cc + r2*v;
79
80 "D Component Mass Balance"
81 v*diff(Cd) = Inlet.Fvol*Inlet.Cd - Outlet.Fvol*Outlet.Cd + r3*v;
```

Criação dos modelos

Modelo de reator de Van der Vusse (equações, 2a parte)


```
83 "Well Mixed - A component"  
84   Outlet.Ca = Ca;  
85  
86 "Well Mixed - B component"  
87   Outlet.Cb = Cb;  
88  
89 "Well Mixed - C component"  
90   Outlet.Cc = Cc;  
91  
92 "Well Mixed - D component"  
93   Outlet.Cd = Cd;  
94  
95 "Residence Time"  
96   tau = V/Inlet.Fvol;  
97  
98 "Flow Equation"  
99   Outlet.Fvol = Inlet.Fvol;  
100  
101 "Thermal equilibrium"  
102   Outlet.T = Inlet.T;  
103  
104 "Mechanical equilibrium"  
105   Outlet.P = Inlet.P;  
106  
107 end
```


Criação da simulação

```
109 ▼FlowSheet VanDerVusse
110
111 DEVICES
112   feed      as MaterialSource (Brief = "Feed stream");
113   reactor   as cstr_vdv       (Brief = "Cstr reactor");
114   product   as MaterialSink   (Brief = "Product");
115
116 CONNECTIONS
117   feed.Outlet to reactor.Inlet;
118   reactor.Outlet to product.Inlet;
119
120 SET
121   reactor.k1 = 0.5 * '1/min';
122   reactor.k2 = 0.2 * '1/min';
123   reactor.k3 = 0.2 * '(1/min)*(1/mol)';
124   reactor.V  = 1000 * 'l';
125
126 SPECIFY
127   feed.Outlet.Ca = 0.05 * 'mol/l';
128   feed.Outlet.Cb = 0 * 'mol/l';
129   feed.Outlet.Cc = 0 * 'mol/l';
130   feed.Outlet.Cd = 0 * 'mol/l';
131   feed.Outlet.Fvol = 250 * 'l/min';
132   feed.Outlet.T = 550 * 'K';
133   feed.Outlet.P = 3 * 'bar';
134
135 INITIAL
136   reactor.Ca = 0.05 * 'mol/l';
137   reactor.Cb = 0 * 'mol/l';
138   reactor.Cc = 0 * 'mol/l';
139   reactor.Cd = 0 * 'mol/l';
140
141 OPTIONS
142   TimeStep = 0.25;
143   TimeEnd = 20;
144   TimeUnit = 'min';
145   Dynamic = true;
146
147 end
```


Resultados da simulação dinâmica

Taxas de reação




Resultados da simulação dinâmica

Concentração do reagente A




Resultados da simulação dinâmica

Concentração do produto de interesse B




Resultados da simulação dinâmica

Concentração do produto não desejado C



Resultados da simulação dinâmica

Concentração do produto não desejado D



Dúvidas?



Recados importantes

- Próxima aula: Apresentação do trabalho final
- Os slides desta aula estarão disponíveis no Classroom da disciplina.

“Ensinar não é transferir conhecimento, mas criar as possibilidades para a sua própria produção ou a sua construção.”

Paulo Freire